0: $\int u \, dx$

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  Int[DeactivateTrig[u,x],x] /;
SimplifyFlag && FunctionOfTrigOfLinearQ[u,x],

Int[u_,x_Symbol] :=
  Int[DeactivateTrig[u,x],x] /;
FunctionOfTrigOfLinearQ[u,x]]
```

Rules for integrands of the form $(a \operatorname{Sin}[e+f x])^m (b \operatorname{Trg}[e+f x])^n$

1. $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx$

1: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx$ when $m+n+2=0 \land m \neq -1$

Reference: G&R 2.510.3, CRC 334a, A&S 4.3.128b with $m+n+2=0$

Reference: G&R 2.510.6, CRC 334b, A&S 4.3.128a with $m+n+2=0$

–

Rule: If $m+n+2=0 \land m \neq -1$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx \longrightarrow \frac{(a \operatorname{Sin}[e+f x])^{m+1} (b \operatorname{Cos}[e+f x])^{n+1}}{a b f (m+1)}$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*(b_.*cos[e_.+f_.*x_])^n_.,x_Symbol] :=
  (a*Sin[e+f*x])^(m+1)*(b*Cos[e+f*x])^(n+1)/(a*b*f*(m+1)) /;
FreeQ[{a,b,e,f,m,n},x] && EqQ[m+n+2,0] && NeQ[m,-1]
```

2: $\int (a \operatorname{Sin}[e+f x])^m \operatorname{Cos}[e+f x]^n \, dx$ when $\frac{n-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2} \in \mathbb{Z}$, then

$$(a \operatorname{Sin}[e+f x])^m \operatorname{Cos}[e+f x]^n == \frac{1}{a f} \operatorname{Subst}\left[x^m \left(1-\frac{x^2}{a^2}\right)^{\frac{n-1}{2}}, x, a \operatorname{Sin}[e+f x]\right] \partial_x (a \operatorname{Sin}[e+f x])$$

Rule: If $\frac{n-1}{2} \in \mathbb{Z}$, then

$$\int (a \operatorname{Sin}[e+f x])^m \operatorname{Cos}[e+f x]^n \, dx \rightarrow \frac{1}{a f} \operatorname{Subst}\left[\int x^m \left(1-\frac{x^2}{a^2}\right)^{\frac{n-1}{2}} dx, x, a \operatorname{Sin}[e+f x]\right]$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*cos[e_.+f_.*x_]^n_.,x_Symbol] :=
  1/(a*f)*Subst[Int[x^m*(1-x^2/a^2)^((n-1)/2),x],x,a*Sin[e+f*x]] /;
FreeQ[{a,e,f,m},x] && IntegerQ[(n-1)/2] && Not[IntegerQ[(m-1)/2] && LtQ[0,m,n]]
```

```
Int[(a_.*cos[e_.+f_.*x_])^m_.*sin[e_.+f_.*x_]^n_.,x_Symbol] :=
  -1/(a*f)*Subst[Int[x^m*(1-x^2/a^2)^((n-1)/2),x],x,a*Cos[e+f*x]] /;
FreeQ[{a,e,f,m},x] && IntegerQ[(n-1)/2] && Not[IntegerQ[(m-1)/2] && GtQ[m,0] && LeQ[m,n]]
```

3. $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx$ when $m > 1$

1: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx$ when $m > 1 \land n < -1$

Reference: G&R 2.510.1

Reference: G&R 2.510.4

Rule: If $m > 1 \land n < -1$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx \longrightarrow$$
$$-\frac{a (a \operatorname{Sin}[e+f x])^{m-1} (b \operatorname{Cos}[e+f x])^{n+1}}{b f (n+1)}+\frac{a^2 (m-1)}{b^2 (n+1)} \int (a \operatorname{Sin}[e+f x])^{m-2} (b \operatorname{Cos}[e+f x])^{n+2} \, dx$$

–

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*cos[e_.+f_.*x_])^n_,x_Symbol] :=
  -a*(a*Sin[e+f*x])^(m-1)*(b*Cos[e+f*x])^(n+1)/(b*f*(n+1)) +
  a^2*(m-1)/(b^2*(n+1))*Int[(a*Sin[e+f*x])^(m-2)*(b*Cos[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f},x] && GtQ[m,1] && LtQ[n,-1] && (IntegersQ[2*m,2*n] || EqQ[m+n,0])
```

```
Int[(a_.*cos[e_.+f_.*x_])^m_*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a*(a*Cos[e+f*x])^(m-1)*(b*Sin[e+f*x])^(n+1)/(b*f*(n+1)) +
  a^2*(m-1)/(b^2*(n+1))*Int[(a*Cos[e+f*x])^(m-2)*(b*Sin[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f},x] && GtQ[m,1] && LtQ[n,-1] && (IntegersQ[2*m,2*n] || EqQ[m+n,0])
```

2: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx$ when $m > 1 \land m+n \neq 0$

Reference: G&R 2.510.2, CRC 323b, A&S 4.3.127b

Reference: G&R 2.510.5, CRC 323a, A&S 4.3.127a

Rule: If $m > 1 \land m+n \neq 0$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx \longrightarrow$$
$$-\frac{a (a \operatorname{Sin}[e+f x])^{m-1} (b \operatorname{Cos}[e+f x])^{n+1}}{b f (m+n)} + \frac{a^2 (m-1)}{m+n} \int (a \operatorname{Sin}[e+f x])^{m-2} (b \operatorname{Cos}[e+f x])^n \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*cos[e_.+f_.*x_])^n_,x_Symbol] :=
  -a*(b*Cos[e+f*x])^(n+1)*(a*Sin[e+f*x])^(m-1)/(b*f*(m+n)) +
  a^2*(m-1)/(m+n)*Int[(b*Cos[e+f*x])^n*(a*Sin[e+f*x])^(m-2),x] /;
FreeQ[{a,b,e,f,n},x] && GtQ[m,1] && NeQ[m+n,0] && IntegersQ[2*m,2*n]
```

```
Int[(a_.*cos[e_.+f_.*x_])^m_*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a*(b*Sin[e+f*x])^(n+1)*(a*Cos[e+f*x])^(m-1)/(b*f*(m+n)) +
  a^2*(m-1)/(m+n)*Int[(b*Sin[e+f*x])^n*(a*Cos[e+f*x])^(m-2),x] /;
FreeQ[{a,b,e,f,n},x] && GtQ[m,1] && NeQ[m+n,0] && IntegersQ[2*m,2*n]
```

4: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx$ when $m < -1$

Reference: G&R 2.510.3, CRC 334a, A&S 4.3.128b

Reference: G&R 2.510.6, CRC 334b, A&S 4.3.128a

Rule: If $m < -1$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx \longrightarrow$$
$$\frac{(a \operatorname{Sin}[e+f x])^{m+1} (b \operatorname{Cos}[e+f x])^{n+1}}{a b f (m+1)} + \frac{m+n+2}{a^2 (m+1)} \int (a \operatorname{Sin}[e+f x])^{m+2} (b \operatorname{Cos}[e+f x])^n \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*cos[e_.+f_.*x_])^n_,x_Symbol] :=
  (b*Cos[e+f*x])^(n+1)*(a*Sin[e+f*x])^(m+1)/(a*b*f*(m+1)) +
  (m+n+2)/(a^2*(m+1))*Int[(b*Cos[e+f*x])^n*(a*Sin[e+f*x])^(m+2),x] /;
FreeQ[{a,b,e,f,n},x] && LtQ[m,-1] && IntegersQ[2*m,2*n]
```

```
Int[(a_.*cos[e_.+f_.*x_])^m_*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -(b*Sin[e+f*x])^(n+1)*(a*Cos[e+f*x])^(m+1)/(a*b*f*(m+1)) +
  (m+n+2)/(a^2*(m+1))*Int[(b*Sin[e+f*x])^n*(a*Cos[e+f*x])^(m+2),x] /;
FreeQ[{a,b,e,f,n},x] && LtQ[m,-1] && IntegersQ[2*m,2*n]
```

5: $\int \sqrt{a \operatorname{Sin}[e+f x]} \sqrt{b \operatorname{Cos}[e+f x]} \, dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a \operatorname{Sin}[e+f x]} \sqrt{b \operatorname{Cos}[e+f x]}}{\sqrt{\operatorname{Sin}[2 e+2 f x]}} == 0$

Rule:

$$\int \sqrt{a \operatorname{Sin}[e+f x]} \sqrt{b \operatorname{Cos}[e+f x]} \, dx \longrightarrow \frac{\sqrt{a \operatorname{Sin}[e+f x]} \sqrt{b \operatorname{Cos}[e+f x]}}{\sqrt{\operatorname{Sin}[2 e+2 f x]}} \int \sqrt{\operatorname{Sin}[2 e+2 f x]} \, dx$$

Program code:

```
Int[Sqrt[a_.*sin[e_.+f_.*x_]]*Sqrt[b_.*cos[e_.+f_.*x_]],x_Symbol] :=
  Sqrt[a*Sin[e+f*x]]*Sqrt[b*Cos[e+f*x]]/Sqrt[Sin[2*e+2*f*x]]*Int[Sqrt[Sin[2*e+2*f*x]],x] /;
FreeQ[{a,b,e,f},x]
```

6: $\int \frac{1}{\sqrt{a \operatorname{Sin}[e+f x]} \sqrt{b \operatorname{Cos}[e+f x]}} \, dx$

- Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{\sqrt{\operatorname{Sin}[2 e+2 f x]}}{\sqrt{a \operatorname{Sin}[e+f x]} \sqrt{b \operatorname{Cos}[e+f x]}} == 0$

- Rule:

$$\int \frac{1}{\sqrt{a \operatorname{Sin}[e+f x]} \sqrt{b \operatorname{Cos}[e+f x]}} \, dx \to \frac{\sqrt{\operatorname{Sin}[2 e+2 f x]}}{\sqrt{a \operatorname{Sin}[e+f x]} \sqrt{b \operatorname{Cos}[e+f x]}} \int \frac{1}{\sqrt{\operatorname{Sin}[2 e+2 f x]}} \, dx$$

- Program code:

```
Int[1/(Sqrt[a_.*sin[e_.+f_.*x_]]*Sqrt[b_.*cos[e_.+f_.*x_]]),x_Symbol] :=
  Sqrt[Sin[2*e+2*f*x]]/(Sqrt[a*Sin[e+f*x]]*Sqrt[b*Cos[e+f*x]])*Int[1/Sqrt[Sin[2*e+2*f*x]],x] /;
FreeQ[{a,b,e,f},x]
```

x: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx$ when $m+n == 0$

Derivation: Piecewise constant extraction

Basis: If $m+n == 0$, then $\partial_x \frac{(a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n}{(a \operatorname{Tan}[e+f x])^m} == 0$

Rule: If $m+n == 0$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx \longrightarrow \frac{(a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n}{(a \operatorname{Tan}[e+f x])^m} \int (a \operatorname{Tan}[e+f x])^m \, dx$$

Program code:

```
(* Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*cos[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Sin[e+f*x])^m*(b*Cos[e+f*x])^n/(a*Tan[e+f*x])^m*Int[(a*Tan[e+f*x])^m,x] /;
FreeQ[{a,b,e,f,m,n},x] && EqQ[m+n,0] *)
```

7: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx$ when $m+n == 0 \land 0 < m < 1$

Derivation: Integration by substitution

Basis: If $-1 < m < 1$, let $k \to \text{Denominator}[m]$, then

$$\frac{(a \operatorname{Sin}[e+f x])^m}{(b \operatorname{Cos}[e+f x])^m} == \frac{k\,a\,b}{f} \operatorname{Subst}\left[\frac{x^{k\,(m+1)-1}}{a^2+b^2 x^{2k}},\ x,\ \frac{(a \operatorname{Sin}[e+f x])^{1/k}}{(b \operatorname{Cos}[e+f x])^{1/k}}\right] \partial_x \frac{(a \operatorname{Sin}[e+f x])^{1/k}}{(b \operatorname{Cos}[e+f x])^{1/k}}$$

Note: This rule is analogous to the rule for integrands of the form $(a \operatorname{Tan}[e+f x])^m$ when $-1 < m < 1$.

Rule: If $m+n == 0 \land 0 < m < 1$, let $k \to \text{Denominator}[m]$, then

$$\int \frac{(a \operatorname{Sin}[e+f x])^m}{(b \operatorname{Cos}[e+f x])^m} \, dx \longrightarrow \frac{k\,a\,b}{f} \operatorname{Subst}\left[\int \frac{x^{k\,(m+1)-1}}{a^2+b^2 x^{2k}} \, dx,\ x,\ \frac{(a \operatorname{Sin}[e+f x])^{1/k}}{(b \operatorname{Cos}[e+f x])^{1/k}}\right]$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*cos[e_.+f_.*x_])^n_,x_Symbol] :=
  With[{k=Denominator[m]},
  k*a*b/f*Subst[Int[x^(k*(m+1)-1)/(a^2+b^2*x^(2*k)),x],x,(a*Sin[e+f*x])^(1/k)/(b*Cos[e+f*x])^(1/k)]] /;
FreeQ[{a,b,e,f},x] && EqQ[m+n,0] && GtQ[m,0] && LtQ[m,1]
```

```
Int[(a_.*cos[e_.+f_.*x_])^m_*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  With[{k=Denominator[m]},
  -k*a*b/f*Subst[Int[x^(k*(m+1)-1)/(a^2+b^2*x^(2*k)),x],x,(a*Cos[e+f*x])^(1/k)/(b*Sin[e+f*x])^(1/k)]] /;
FreeQ[{a,b,e,f},x] && EqQ[m+n,0] && GtQ[m,0] && LtQ[m,1]
```

8: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx$

- Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{(b \operatorname{Cos}[e+f x])^{n-1}}{\left(\operatorname{Cos}[e+f x]^2\right)^{\frac{n-1}{2}}} == 0$

Basis: $\operatorname{Cos}[e+f x] \, F[a \operatorname{Sin}[e+f x]] == \frac{1}{a f} \operatorname{Subst}[F[x], x, a \operatorname{Sin}[e+f x]] \, \partial_x (a \operatorname{Sin}[e+f x])$

Note: If $\frac{n}{2} \in \mathbb{Z} \land 3 m \in \mathbb{Z} \land -1 < m < 1$, integration of $x^m \left(1-\frac{x^2}{a^2}\right)^{\frac{n-1}{2}}$ results in a complicated antiderivative involving elliptic integrals and the imaginary unit.

- Rule:

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Cos}[e+f x])^n \, dx \to \frac{b^{2 \operatorname{IntPart}\left[\frac{n-1}{2}\right]+1} (b \operatorname{Cos}[e+f x])^{2 \operatorname{FracPart}\left[\frac{n-1}{2}\right]}}{\left(\operatorname{Cos}[e+f x]^2\right)^{\operatorname{FracPart}\left[\frac{n-1}{2}\right]}} \int \operatorname{Cos}[e+f x] \, (a \operatorname{Sin}[e+f x])^m \left(1-\operatorname{Sin}[e+f x]^2\right)^{\frac{n-1}{2}} dx$$

$$\to \frac{b^{2 \operatorname{IntPart}\left[\frac{n-1}{2}\right]+1} (b \operatorname{Cos}[e+f x])^{2 \operatorname{FracPart}\left[\frac{n-1}{2}\right]}}{a f \left(\operatorname{Cos}[e+f x]^2\right)^{\operatorname{FracPart}\left[\frac{n-1}{2}\right]}} \operatorname{Subst}\left[\int x^m \left(1-\frac{x^2}{a^2}\right)^{\frac{n-1}{2}} dx, x, a \operatorname{Sin}[e+f x]\right]$$

$$\to \frac{b^{2 \operatorname{IntPart}\left[\frac{n-1}{2}\right]+1} (b \operatorname{Cos}[e+f x])^{2 \operatorname{FracPart}\left[\frac{n-1}{2}\right]} (a \operatorname{Sin}[e+f x])^{m+1}}{a f (m+1) \left(\operatorname{Cos}[e+f x]^2\right)^{\operatorname{FracPart}\left[\frac{n-1}{2}\right]}} \operatorname{Hypergeometric2F1}\left[\frac{1+m}{2}, \frac{1-n}{2}, \frac{3+m}{2}, \operatorname{Sin}[e+f x]^2\right]$$

- Program code:

```
(* Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*cos[e_.+f_.*x_])^n_,x_Symbol] :=
  b^(2*IntPart[(n-1)/2]+1)*(b*Cos[e+f*x])^(2*FracPart[(n-1)/2])/(a*f*(Cos[e+f*x]^2)^FracPart[(n-1)/2])*
    Subst[Int[x^m*(1-x^2/a^2)^((n-1)/2),x],x,a*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m,n},x] && (RationalQ[n] || Not[RationalQ[m]] && (EqQ[b,1] || NeQ[a,1])) *)
```

```
(* Int[(a_.*cos[e_.+f_.*x_])^m_*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^(2*IntPart[(n-1)/2]+1)*(b*Sin[e+f*x])^(2*FracPart[(n-1)/2])/(a*f*(Sin[e+f*x]^2)^FracPart[(n-1)/2])*
    Subst[Int[x^m*(1-x^2/a^2)^((n-1)/2),x],x,a*Cos[e+f*x]] /;
FreeQ[{a,b,e,f,m,n},x] *)
```

```
Int[(a_.*cos[e_.+f_.*x_])^m_*(b_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^(2*IntPart[(n-1)/2]+1)*(b*Sin[e+f*x])^(2*FracPart[(n-1)/2])*(a*Cos[e+f*x])^(m+1)/(a*f*(m+1)*(Sin[e+f*x]^2)^FracPart[(n-1)/2])*
    Hypergeometric2F1[(1+m)/2,(1-n)/2,(3+m)/2,Cos[e+f*x]^2] /;
FreeQ[{a,b,e,f,m,n},x] && SimplerQ[n,m]
```

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*cos[e_.+f_.*x_])^n_,x_Symbol] :=
  b^(2*IntPart[(n-1)/2]+1)*(b*Cos[e+f*x])^(2*FracPart[(n-1)/2])*(a*Sin[e+f*x])^(m+1)/(a*f*(m+1)*(Cos[e+f*x]^2)^FracPart[(n-1)/2])*
    Hypergeometric2F1[(1+m)/2,(1-n)/2,(3+m)/2,Sin[e+f*x]^2] /;
FreeQ[{a,b,e,f,m,n},x]
```

2. $\int (a\,\mathrm{Sin}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $m\notin\mathbb{Z}\land n\notin\mathbb{Z}$

1: $\int (a\,\mathrm{Sin}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $m-n+2==0\land m\neq -1$

Rule: If $m-n+2==0\land m\neq -1$, then

$$\int (a\,\mathrm{Sin}[e+f\,x])^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx \longrightarrow \frac{b\,(a\,\mathrm{Sin}[e+f\,x])^{m+1}\,(b\,\mathrm{Sec}[e+f\,x])^{n-1}}{a\,f\,(m+1)}$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*(b_.*sec[e_.+f_.*x_])^n_.,x_Symbol] :=
  b*(a*Sin[e+f*x])^(m+1)*(b*Sec[e+f*x])^(n-1)/(a*f*(m+1)) /;
FreeQ[{a,b,e,f,m,n},x] && EqQ[m-n+2,0] && NeQ[m,-1]
```

2. $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $n > 1$

1: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $n > 1 \land m > 1$

Rule: If $n > 1 \land m > 1$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx \longrightarrow$$
$$\frac{a b (a \operatorname{Sin}[e+f x])^{m-1} (b \operatorname{Sec}[e+f x])^{n-1}}{f (n-1)} - \frac{a^2 b^2 (m-1)}{n-1} \int (a \operatorname{Sin}[e+f x])^{m-2} (b \operatorname{Sec}[e+f x])^{n-2} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  a*b*(a*Sin[e+f*x])^(m-1)*(b*Sec[e+f*x])^(n-1)/(f*(n-1)) -
  a^2*b^2*(m-1)/(n-1)*Int[(a*Sin[e+f*x])^(m-2)*(b*Sec[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f},x] && GtQ[n,1] && GtQ[m,1] && IntegersQ[2*m,2*n]
```

2: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $n > 1$

Rule: If $n > 1$, then

$$\frac{n-1}{b^2 (m-n+2)} \int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx \longrightarrow$$
$$\frac{b (a \operatorname{Sin}[e+f x])^{m+1} (b \operatorname{Sec}[e+f x])^{n-1}}{a f (n-1)} - \frac{b^2 (m-n+2)}{n-1} \int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^{n-2} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Sin[e+f*x])^(m+1)*(b*Sec[e+f*x])^(n+1)/(a*b*f*(m-n)) -
  (n+1)/(b^2*(m-n))*Int[(a*Sin[e+f*x])^m*(b*Sec[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f,m},x] && GtQ[n,1] && IntegersQ[2*m,2*n]
```

3. $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $n < -1$

1: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $n < -1 \land m < -1$

Rule: If $n < -1 \land m < -1$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx \longrightarrow$$
$$\frac{(a \operatorname{Sin}[e+f x])^{m+1} (b \operatorname{Sec}[e+f x])^{n+1}}{a b f (m+1)} - \frac{n+1}{a^2 b^2 (m+1)} \int (a \operatorname{Sin}[e+f x])^{m+2} (b \operatorname{Sec}[e+f x])^{n+2} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Sin[e+f*x])^(m+1)*(b*Sec[e+f*x])^(n+1)/(a*b*f*(m+1)) -
  (n+1)/(a^2*b^2*(m+1))*Int[(a*Sin[e+f*x])^(m+2)*(b*Sec[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f},x] && LtQ[n,-1] && LtQ[m,-1] && IntegersQ[2*m,2*n]
```

2: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $n < -1 \land m - n \neq 0$

Rule: If $n < -1 \land m - n \neq 0$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx \to$$
$$\frac{(a \operatorname{Sin}[e+f x])^{m+1} (b \operatorname{Sec}[e+f x])^{n+1}}{a b f (m-n)} - \frac{n+1}{b^2 (m-n)} \int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^{n+2} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Sin[e+f*x])^(m+1)*(b*Sec[e+f*x])^(n+1)/(a*b*f*(m-n)) -
  (n+1)/(b^2*(m-n))*Int[(a*Sin[e+f*x])^m*(b*Sec[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f,m},x] && LtQ[n,-1] && NeQ[m-n,0] && IntegersQ[2*m,2*n]
```

4: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $m > 1 \land m - n \neq 0$

Rule: If $m > 1 \land m - n \neq 0$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx \to$$
$$-\frac{a b (a \operatorname{Sin}[e+f x])^{m-1} (b \operatorname{Sec}[e+f x])^{n-1}}{f (m-n)} + \frac{a^2 (m-1)}{m-n} \int (a \operatorname{Sin}[e+f x])^{m-2} (b \operatorname{Sec}[e+f x])^n \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  -a*b*(a*Sin[e+f*x])^(m-1)*(b*Sec[e+f*x])^(n-1)/(f*(m-n)) +
  a^2*(m-1)/(m-n)*Int[(a*Sin[e+f*x])^(m-2)*(b*Sec[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && GtQ[m,1] && NeQ[m-n,0] && IntegersQ[2*m,2*n]
```

5: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $m < -1$

Rule: If $m < -1$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx \longrightarrow$$
$$\frac{b (a \operatorname{Sin}[e+f x])^{m+1} (b \operatorname{Sec}[e+f x])^{n-1}}{a f (m+1)} + \frac{m-n+2}{a^2 (m+1)} \int (a \operatorname{Sin}[e+f x])^{m+2} (b \operatorname{Sec}[e+f x])^n \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a*Sin[e+f*x])^(m+1)*(b*Sec[e+f*x])^(n-1)/(a*f*(m+1)) +
  (m-n+2)/(a^2*(m+1))*Int[(a*Sin[e+f*x])^(m+2)*(b*Sec[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && LtQ[m,-1] && IntegersQ[2*m,2*n]
```

6. $\int (a \sin[e+f x])^m (b \sec[e+f x])^n \, dx$ when $m \notin \mathbb{Z} \land n \notin \mathbb{Z}$

1: $\int (a \sin[e+f x])^m (b \sec[e+f x])^n \, dx$ when $m-\frac{1}{2} \in \mathbb{Z} \land n-\frac{1}{2} \in \mathbb{Z}$

- Derivation: Piecewise constant extraction

- Basis: $\partial_x \left((b \cos[e+f x])^n (b \sec[e+f x])^n\right) = 0$

- Rule: If $m-\frac{1}{2} \in \mathbb{Z} \land n-\frac{1}{2} \in \mathbb{Z}$, then

$$\int (a \sin[e+f x])^m (b \sec[e+f x])^n \, dx \longrightarrow (b \cos[e+f x])^n (b \sec[e+f x])^n \int \frac{(a \sin[e+f x])^m}{(b \cos[e+f x])^n} \, dx$$

- Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  (b*Cos[e+f*x])^n*(b*Sec[e+f*x])^n*Int[(a*Sin[e+f*x])^m/(b*Cos[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,m,n},x] && IntegerQ[m-1/2] && IntegerQ[n-1/2]
```

2: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $m \notin \mathbb{Z} \land n \notin \mathbb{Z} \land n < 1$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((b \operatorname{Cos}[e+f x])^{n+1} (b \operatorname{Sec}[e+f x])^{n+1} \right) == 0$

Rule: If $m \notin \mathbb{Z} \land n \notin \mathbb{Z} \land n < 1$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Sec}[e+f x])^n \, dx \longrightarrow \frac{1}{b^2} (b \operatorname{Cos}[e+f x])^{n+1} (b \operatorname{Sec}[e+f x])^{n+1} \int \frac{(a \operatorname{Sin}[e+f x])^m}{(b \operatorname{Cos}[e+f x])^n} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  1/b^2*(b*Cos[e+f*x])^(n+1)*(b*Sec[e+f*x])^(n+1)*Int[(a*Sin[e+f*x])^m/(b*Cos[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && LtQ[n,1]
```

3: $\int (a \sin[e+f x])^m (b \sec[e+f x])^n \, dx$ when $m \notin \mathbb{Z} \land n \notin \mathbb{Z}$

- Derivation: Piecewise constant extraction

- Basis: $\partial_x \left((b \cos[e+f x])^{n-1} (b \sec[e+f x])^{n-1} \right) == 0$

Rule: If $m \notin \mathbb{Z} \land n \notin \mathbb{Z}$, then

$$\int (a \sin[e+f x])^m (b \sec[e+f x])^n \, dx \longrightarrow b^2 (b \cos[e+f x])^{n-1} (b \sec[e+f x])^{n-1} \int \frac{(a \sin[e+f x])^m}{(b \cos[e+f x])^n} \, dx$$

- Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*sec[e_.+f_.*x_])^n_,x_Symbol] :=
  b^2*(b*Cos[e+f*x])^(n-1)*(b*Sec[e+f*x])^(n-1)*Int[(a*Sin[e+f*x])^m/(b*Cos[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```

3: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Csc}[e+f x])^n \, dx$ when $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((a \operatorname{Sin}[e+f x])^n (b \operatorname{Csc}[e+f x])^n\right) == 0$

Rule: If $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Csc}[e+f x])^n \, dx \longrightarrow (a b)^{\text{IntPart}[n]} (a \operatorname{Sin}[e+f x])^{\text{FracPart}[n]} (b \operatorname{Csc}[e+f x])^{\text{FracPart}[n]} \int (a \operatorname{Sin}[e+f x])^{m-n} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*(b_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*b)^IntPart[n]*(a*Sin[e+f*x])^FracPart[n]*(b*Csc[e+f*x])^FracPart[n]*Int[(a*Sin[e+f*x])^(m-n),x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```